
An important problem with many useful applications is finding a mathematical model that replicates certain features of an experimental data record.  The model might be useful for prediction, for generation of additional data, for numerical experimentation, or for illuminating the underlying dynamics.  The choice of an appropriate model and the method used to fit the model to the data are strongly influenced by the purpose for which the model is to be used.  For example, a model optimized for good short-term prediction will usually not do well at capturing the long-term dynamics, and vice versa, especially if the data come from chaotic or noisy dynamics.


This note will describe a numerical algorithm that finds a model equation whose chaotic solutions replicate certain features of the experimental data.  The feature chosen is the power spectrum, or more particularly, the auto-correlation function, whose Fourier transform is the power spectrum.  Since chaotic data will usually have a broadband power spectrum, the method nearly always produces a model with chaotic solutions, since non-chaotic models would have discrete power spectra.  

The model is generally not useful for prediction since it does not preserve the underlying dynamics.  In fact, a common technique for generating surrogate data with all the determinism removed is to Fourier transform the data, randomize the phases, and inverse Fourier transform the result, leaving the power spectrum unchanged.  However, the method might allow prediction if the model is sufficiently constrained by other factors.  It might also be used to exclude certain models by showing that they are incapable of replicating the power spectrum of the data.  There may be other applications such as the rapid generation of an unlimited time series with certain spectral properties for testing filters or the bandpass characteristics of communications channels.

The method consists of calculating the auto-correlation function of the data for lags in the range of 1 to 50 (corresponding to frequencies from the Nyquist frequency, 0.5/T, to 0.01/T, where T is the sample time).  The same quantity is calculated for the model equation, whose coefficients are then adjusted to give the least square fit.  The optimization method is a variant of simulated annealing in which the initial coefficients are randomly chosen, and then neighborhoods in the coefficient space are randomly explored.  The size of the neighborhood is initially large but gradually shrinks.  Whenever an improved solution is found, its neighborhood is expanded and similarly explored.  Both the data and the model are adjusted to have mean 0 and variance 1 before comparisons are made.  Initial conditions for the model equation are chosen at 0, and the first 1000 points of the solution are discarded to help ensure that the solution is on the attractor.  

Appendix I shows an implementation of the method in BASIC.  BASIC was chosen for readability and because when compiled with PowerBASIC for DOS, execution is as fast or faster than with other languages.  The program will also run under the much slower Microsoft QuickBASIC.  A relatively simple model equation was used of the form

xn+1 = a1 + a2xn + a3xn2 + a4xnxn-1 + a5xn-1 + a6xn-12

This model is easily generalized to higher dimensions (more time lags) or polynomial degree, but it is surprisingly general with appropriate choices of the six coefficients, a1 through a6.  The program inputs data from an ASCII file CHAOSFIT.DAT, and it outputs a time series of the same length from the model equations to the file CHAOSFIT.FIT.  It will run forever, constantly seeking improved solutions, or until the <Esc> key is pressed.


An example of the screen output from the program is shown below for a case in which the input data is 2000 points of Gaussian 1/f noise.

The numbers at the top are the optimized values of the six coefficients.  The upper trace is the first 50 values of the input data, and the middle trace is the first 50 values of the output data.  The two traces at the bottom that almost perfectly overlay are the auto-correlation function of the data and the model solution, respectively. The solutions obtained by this method are usually not unique (other very different values of the coefficients can give similar fits), but they are usually robust to small variations of the coefficients.

Appendix I

'Program CHAOSFIT.BAS fits a chaotic model to experimental data

'(c)1997 by J. C. Sprott (http://sprott.physics.wisc.edu/)

SCREEN 12

jmax% = 50  'Maximum time to correlate

ip% = 50    'Number of time points to plot

kmax% = 6   'Number of coefficients

dmax% = 2   'Number of variables (embedding dimension)

DIM cd(jmax%), cx(jmax%), y(dmax%), a(kmax%), am(kmax%)

OPEN "CHAOSFIT.DAT" FOR INPUT AS #1     'Get length of input data record

imax% = 0

WHILE NOT EOF(1)

    imax% = imax% + 1

    INPUT #1, d

WEND

CLOSE #1

imax% = imax% - 1

DIM d(imax%), x(imax%)

OPEN "CHAOSFIT.DAT" FOR INPUT AS #1     'Read input data

dav = 0

FOR i% = 0 TO imax%

    INPUT #1, d(i%)

    dav = dav + d(i%)

NEXT i%

dav = dav / (imax% + 1)     'Average value of input data

CLOSE #1

FOR j% = 0 TO jmax%         'Calculate autocorrelation of input data

    cd(j%) = 0

    FOR i% = 0 TO imax% - j%

        cd(j%) = cd(j%) + (d(i%) - dav) * (d(i% + j%) - dav)

    NEXT i%

    IF j% = 0 THEN cd0 = cd(0)

    cd(j%) = cd(j%) / cd0

NEXT j%

cd0 = SQR(cd0 / (imax% + 1))    'RMS value of input data

FOR i% = 0 TO imax%: d(i%) = (d(i%) - dav) / cd0: NEXT i%'Normalize to cd0

RANDOMIZE TIMER

rmin = 1E+38: eps = 4: rbest = rmin

WHILE INKEY$ <> CHR$(27)        'Loop until the <Esc> key is pressed

    FOR k% = 1 TO kmax%: a(k%) = am(k%) + eps * (RND - .5): NEXT k%

    FOR j% = 1 TO dmax%: y(j%) = 0: NEXT j% 'Zero initial conditions

    xav = 0

    FOR i% = 1 TO 1000 + imax%  'Generate model equation data

        IF ABS(y(1)) > 100 THEN EXIT FOR    'unbounded

        y(0) = a(1) + a(2) * y(1) + a(3) * y(1) * y(1) + a(4) * y(1) * y(2) + a(5) * y(2) + a(6) * y(2) * y(2)

        IF y(0) = y(1) THEN EXIT FOR        'fixed point

        FOR j% = dmax% TO 1 STEP -1: y(j%) = y(j% - 1): NEXT j%

        IF i% > 999 THEN x(i% - 1000) = y(1): xav = xav + y(1)

    NEXT i%

    IF i% = imax% + 1001 THEN   'Calculate autocorrelation of model data

        xav = xav / (imax% + 1)

        FOR i% = 0 TO imax%: x(i%) = x(i%) - xav: NEXT i%   'Subtract average

        FOR j% = 0 TO jmax%

            cx(j%) = 0

            FOR i% = 0 TO imax% - j%

                cx(j%) = cx(j%) + x(i%) * x(i% + j%)

            NEXT i%

            IF j% = 0 THEN cx0 = cx(0)

            cx(j%) = cx(j%) / cx0

        NEXT j%

        cx0 = SQR(cx0 / (imax% + 1))    'RMS value of model data

        FOR i% = 0 TO imax%: x(i%) = x(i%) / cx0: NEXT i%   'Normalize to cx0

        r = 0

        FOR j% = 1 TO jmax%     'Calculate sum of squared errors (r)

            rsqrt = cd(j%) - cx(j%)

            r = r + rsqrt * rsqrt

        NEXT j%

        IF r <= rmin AND r > 0 THEN     'Test quality of solution

            rmin = r

            IF rmin < rbest THEN        'A better solution was found

                BEEP

                rbest = rmin

                CLS                     'Graph it

                LINE (0, 100)-(639, 100), 2

                PSET (0, 100 - 40 * d(0))

                FOR i% = 1 TO ip%

                    LINE -(640 * i% / ip%, 100 - 40 * d(i%)), 15

                NEXT i%

                LINE (0, 240)-(639, 240), 2

                PSET (0, 240 - 40 * x(0))

                FOR i% = 1 TO ip%

                    LINE -(640 * i% / ip%, 240 - 40 * x(i%)), 12

                NEXT i%

                LINE (0, 400)-(639, 400), 2

                PSET (0, 300)

                FOR j% = 1 TO jmax%

                    LINE -(640 * j% / jmax%, 400 - 100 * cd(j%)), 15

                NEXT j%

                PSET (0, 300)

                FOR j% = 1 TO jmax%

                    LINE -(640 * j% / jmax%, 400 - 100 * cx(j%)), 12

                NEXT j%

                LOCATE 1, 1

                FOR k% = 1 TO kmax%: PRINT a(k%), ; : NEXT k%

                LINE (0, 0)-(639, 479), , B

            END IF

            FOR k% = 1 TO kmax%: am(k%) = a(k%): NEXT k%    'Move coefficients

            IF eps < 1 THEN eps = 4 * eps

            OPEN "CHAOSFIT.FIT" FOR OUTPUT AS #1            'Save model data

                FOR i% = 0 TO imax%: PRINT #1, x(i%): NEXT i%

            CLOSE #1

        ELSE

            eps = eps / 1.001   'Shrink the search neighborhood (gradually)

            IF eps < .000001 THEN   'Neighborhood too small, start over

                rmin = 1E+38

                eps = 4

                FOR k% = 1 TO kmax%: am(k%) = 0: NEXT k%

            END IF

        END IF

    END IF

WEND

